

المصفوفات

من الصفـر حتـى الاحـتـراف

تأليف:

الاندرتيكر Under Tiker

مشرف منتدى الفيـجـوال بيسـك بـمـنـتـديـات المـحـيط العـربـي
جـمـيـع الحـقـوق مـحـفـوظـة لـمـنـتـدي المـحـيط العـربـي

www.arabmoheet.net

الفهرس :

المقدمة

الاعلان عن المصفوفات

المصفوفات ذات البعد الواحد

الفرق بين المصفوفات الثابتة والديناميكية

المصفوفات متعددة الأبعاد

مهارات في التعامل مع المصفوفات

أمثلة على المصفوفات

المثال الأول (تخزين البيانات داخل مصفوفة ثم ترتيبها وعرضها)

المثال الثاني (تعامل المصفوفات مع قواعد البيانات وإجراء العمليات عليها)

أسئلة الكتاب

المقدمة :

المصفوفة (Array) هي مجموعة من العناصر (المتغيرات) لها اسم واحد و لها نوع بياني واحد ولكل عنصر (متغير) داخل المصفوفة دليل Index فريد يحدد ترتيبه داخل المصفوفة لتمييز كل عنصر عن الآخر. وعند استدعاء أي متغير فيها يتم الإشارة إليه باستخدام الدليل (Index).

و (الفهرس Index) بتعريف مبسط هو الرقم التسلسلي الخاص بكل عنصر داخل المصفوفة.

خذ هذا المثال لتوضح معنى المصفوفة:

لو دخلت على إحدى المكتبات تجد هناك كتب تنقسم إلى عدة مجلدات.. نأخذ مثلا كتاب اسمه (الشعر في العصر الحديث) تجد هذا الكتاب ينقسم إلى ستة مجلدات هنا نقول أن جميع الكتب لها نفس الاسم وهو (الشعر في العصر الحديث) ولكن كل مجلد يختلف عن الآخر برقم فريد مميز فنقول المجلد 1 و المجلد 2 والمجلد 3 .. وهكذا .

الآن أتضح لنا أن المصفوفة لها اسم واحد ونوع بياني واحد ولكن تتكون من عدة عناصر كل عنصر له رقم فريد خاص به يسمى Index.

ملاحظة:

الفهرس (Index) يستخدم للتمييز بين العناصر المختلفة في المصفوفة.

متى نستخدم المصفوفات؟

تستخدم المصفوفات في أمور عديدة منها على سبيل المثال لا الحصر عندما ترغب في حفظ قائمة من القيم مؤقتا لكي يتم الرجوع إليها أثناء عمل البرنامج، ومثلا لو أردت حساب نسبة النجاح والمتوسط وأعلى وأقل درجة لـ 100 طالب ففي هذه الحالة توفر لك المصفوفة الوقت بدلا من الإعلان عن أكثر من 200 متغير وبالتالي فالمتغيرات الفردية (العادية) لا تفي بالغرض هنا فيجب اللجوء إلى استخدام المصفوفات.

من مميزاتها:

- 1- سهولة الاستخدام
- 2- إمكانية استخدام حلقات التكرار للمرور على عناصر المصفوفة.
- 3- تتسع لاحتواء كميات كبيرة من البيانات.
- 4- يمكن زيادة حجم المصفوفة والتحكم بحجمها أثناء عمل البرنامج وهذه الميزة تنفرد بها لغة الـ Visual Basic عن اللغات الأخرى وسوف نتعرف على ذلك لاحقا.

و هناك نوعان المصفوفات:

- 1- مصفوفة ذات بعد واحد Single Dimensional Array
- 2- مصفوفات متعددة الأبعاد Multi Dimensional Arrays

وسوف يتم الحديث عنها بالتفصيل ولكن لمحة بسيطة عنهما نقول أن المصفوفة ذات البعد الواحد تشبه السجل والمصفوفة متعددة الأبعاد تشبه الجدول تقريبا.

الإعلان عن المصفوفات

أولاً:

المصفوفات ذات البعد الواحد:

وهي في الحقيقة تشبه السجل والواحد تكون ذات بعد واحد أفقياً أو رأسياً.

عبارة الإعلان:

عبارة الإعلان عن المصفوفة تشبه عبارة الإعلان عن المتغير العادي.

الشكل العام:

Dim ArrName(Len) AS DataType

وتبدأ بنوع المدى (المجال SCOPE) للمصفوفة:

وهي إما أن تكون Dim أو Public أو Private

ولو أردنا أن نعرف الفرق بين كل واحد منها فنقول:

1- Private : تجعل المصفوفة محلية إذا تم الإعلان عنها في إجراء فإنها لا ترى إلا في هذا الإجراء، وإن تم الإعلان عنها في قسم الإعلان العام في نموذج أو وحدة برمجية فإنها لا ترى إلا في هذا النموذج أو هذه الوحدة.

2- Public : لا تستخدم إلا في قسم الإعلان العام ، وتعني أن المتغير يصبح متاحاً في أي مكان من البرنامج (أي وحدة برمجية أو إجراء أو نموذج)

3- Dim : تشبه Private عند استخدامها في إجراء أو في قسم الإعلان العام في نموذج، بينما تقوم بعمل Public عند استخدامها في قسم الإعلان العام في وحدة برمجية.

ثم تليها ArrName وهي اسم المصفوفة

ثم تحديد طول المصفوفة Len وهذا العدد يحدد عدد العناصر في المصفوفة فمثلاً إن كانت Len=4 فإن طول المصفوفة يساوي 5 لأن الترقيم في المصفوفات يبدأ من الصفر. ولكن هناك طريقة تجعل الترقيم يبدأ بغير الصفر وسوف نعرف ذلك.

ثم في DataType نحدد النوع البياني للمصفوفة.

مثال:

لو كان هنا 50 طالبا ونريد تخزين أسماء الطلاب
فإننا نحتاج لمصفوفة تتكون من 50 عنصرا .. فيكون الإعلان كالتالي:

Dim Stud(49) As String

- اسم المصفوفة Stud
- طولها 50 عنصرا , لان الترقيم في المصفوفات يبدأ من الصفر.
- النوع البياني للمصفوفة هو String

وطريقة التعامل معها كالتالي:

أولا: إضافة البيانات

```
Stud(0) = "Ali Ahmad"
Stud(1) = "Majed Nabil"
Stud(2) = "Saleh Khaled"
Stud(49) = "UnderTaker"
.
```

ملاحظة:

عند وضع قيمة نصية لعنصر داخل المصفوفة فانه يجب احاطتها بـ (" ") Double Quotation

ثانيا الحصول على البيانات:

```
Text1.Text=Stud(20)
Text2.Text=Stud(1)
Text3.Text=Stud(40)
.
```

طرق الإعلان عنها

1- يمكن الإعلان عنها بتحديد عدد العناصر في المصفوفة (القيمة القصوى).

مثال:

Dim ArrName(Len) As DataType

حيث Len تمثل الطول الأقصى للمصفوفة

وفي هذه الحالة يبدأ الترقيم داخل المصفوفة من الصفر حتى الرقم المحدد Len
ولجعل المصفوفة تبدأ الترقيم من الواحد بدلا من الصفر
أعلن عن Option Base في قسم الإعلان العام للوحدة البرمجية :

Option Base 1

وبعد الإعلان عنها لا يمكن التعامل مع العنصر (صفر) لان هذا العنصر ليس له وجود الآن
ولو استخدمته سيتوقف البرنامج مصدرا رسالة خطأ.

ملاحظة:

عند الإعلان عن Option Base فإنه يمكن الترقيم من واحد أو صفر فقط
Option Base 0 أو Option Base 1

مثال:

Dim Arr(2) As String

1 - قبل الإعلان عن Option Base يتم التعامل مع المصفوفة Arr كالتالي:

```
Arr(0) = "AAA"
Arr(1) = "DDD"
Arr(2) = "VVV"
```

2 - وبعد الإعلان عن Option Base يتم التعامل مع المصفوفة Arr كالتالي:

```
Arr(1) = "DDD"
Arr(2) = "VVV"
```

لا حظ لم يتم التعامل مع العنصر صفر .. كما أن طول المصفوفة قد نقص بسبب إلغاء العنصر صفر.

وهذه الطريقة (أي استخدام Option Base) عند كثرة الاكواد واستخدامها كثيرا داخل البرنامج قد توقع بعض الأخطاء أنا لا اجزم ولكن في بعض الأحيان تحدث Error .

ولتفادي هذه المشكلة قم باستخدام الطريقة التالية:

2- يمكن الإعلان عن المصفوفة بتحديد رقم بداية عناصر المصفوفة و نهايتها (طريقة البداية والنهاية).

هذه الطريقة تعتبر أكفأ من استخدام

Option Base 1

و أكثر مرونة لان الطريقة السابقة تجبرك بان تبدأ الترقيم من الواحد فقط أو بالبداية بالقيمة صفر عند عدم الإعلان عن Option Base.

ولكن طريقة (البداية والنهاية) تحدد رقم البداية كما تشاء بدون أن توقع في المشاكل والأخطاء. Error.

مثال 1 :

Dim ArrName(Str_Len TO End_Len) As DataType

حيث Str_Len تمثل بداية المصفوفة.
وحيث End_Len تمثل نهاية المصفوفة.

مثال 2:**Dim XArray(2 To 4) As String**

حيث 2 تمثل بداية المصفوفة.
وحيث 4 تمثل نهاية المصفوفة.

واسم المصفوفة هو XArray
ونوع البياني للمصفوفة هو String

وطول المصفوفة هو 3 , أي تحتوي على ثلاثة عناصر وهي

```
XArray(2) = "AAA"
XArray(3) = "BBB"
XArray(4) = "SSS"
```

3- يمكن الإعلان عنها بدون أن تحتوي على أي عنصر (المصفوفة الخالية من العناصر).

مثال:**Dim ArrName() As DataType**

وفي هذه الحالة تكون المصفوفة فارغة تماما ولا يمكن التعامل معها إلا بعد إعادة تعريفها عن طريق الأمر ReDim وهذا الأمر يقوم بتحديد طول المصفوفة كالتالي:

Dim ArrName() As String**ReDim ArrName(2)**

```
ArrName(0) = "AAA"
ArrName(1) = "BBB"
ArrName(2) = "CCC"
```

وهذه الطريقة لا ننصح باستخدامها بكثرة وسوف يتم الحديث عنها بالتفصيل لاحقا.

4- يمكن الإعلان عن مصفوفة من نوع Variant بوضع القيم باستخدام الدالة Array()

وهذه الدالة لا تعمل إلا مع المصفوفات ذات البعد الواحد.

مثال:

Dim ArrName as Variant

ArrName = Array("Ali","UnderTiker","Mazen","Yaser")

اسم المصفوفة ArrName

طولها: تحتوي على أربعة عناصر
رقم أول عنصر في المصفوفة هو 0
رقم آخر عنصر في المصفوفة هو 3

في حال استخدام Option Base فان

رقم أول عنصر في المصفوفة هو 1
رقم آخر عنصر في المصفوفة هو 4

ما الفرق بين المصفوفات الثابتة والمصفوفات الديناميكية؟

هناك نوعان من تعريف المصفوفات:

1- **المصفوفات الثابتة:** التي تحتوي على عدد عناصر محدد (ثابتة الحجم Fixed Size) أي طوال عمل البرنامج يبقى طولها ثابت.

2- **المصفوفات ديناميكية:** ويطلق عليها اسم Dynamic وهي التي يتغير طول عناصرها فهي مصفوفات يمكن تغيير حجمها أثناء عمل البرنامج ويتم ذلك عن طريق أمر إعادة التعريف للمصفوفة ReDim.

مميزات المصفوفات الديناميكية:

1- عند الإعلان عن مصفوفة عدد عناصرها ثابت يقوم Visual Basic بحجز مساحة من الذاكرة تتسع لجميع عناصر المصفوفة ، فإذا كانت عناصر المصفوفة غير مستغلة كلها فان هذا يعني هدر من مساحة الذاكرة .

2- بعض التطبيقات لا تعرف مقدما عدد العناصر التي ستحتاجها. والحل في هذه الحالة هو استخدام المصفوفة الديناميكية أو المصفوفات متغيرة الحجم.

عيوبها:

1- عند إعادة التعريف قد يتم فقد البيانات او جزء منها.

2- عند كثرة إعادة تعريف مصفوفة قد تفقد السيطرة عليها.

تعتمد فكرة المصفوفة ذات الحجم المتغير أن يتم الإعلان عن مصفوفة خالية من العناصر.. ثم يتم إعادة تعريفها وإضافة عدد مبدئي من العناصر وعند الحاجة لعدد إضافي من العناصر يتم زيادتها بحسب حاجة البرنامج فقط.

ولكي يتم التعامل مع المصفوفات الديناميكية يجب الإعلان مصفوفة فارغة بدون تحديد عدد العناصر كما ذكرنا سابقا (الطريقة الثالثة من طرق الإعلان عن المصفوفة) و إلا فلن يتم السماح بتغيير طولها.

صيغة عبارة ReDim :

ReDim [Preserve] ArrayName(Length) As DataType

ReDim وهي عبارة برمجية وليست إعلانا أي يجب استخدامها داخل إجراء ولا يمكن استخدامها في قسم الإعلان العام.

(Preserve) كلمة اختيارية وتعني الاحتفاظ بالعناصر الموجودة حاليا في المصفوفة. نظرا لأن عبارة ReDim يمكن استخدامها أكثر من مرة مع نفس المصفوفة فقد يتم استخدامها والمصفوفة بها عناصر بالفعل، لو لم يتم وضع Preserve فإنه سوف يتم فقدان القيم الموجودة داخل عناصر المصفوفة، أما إذا تم استخدام Preserve فإن القيم الحالية سيتم الاحتفاظ بها، لو تم زيادة عدد العناصر، أما إذا تم تقليل العدد فإن العبارة ستحتفظ بأكبر عدد ممكن من العناصر (أما العناصر الزائدة عن الطول الجديد للمصفوفة فسوف يتم حذفها)

(ArrayName) اسم المصفوفة.

(Length) طول المصفوفة الجديد.

(DataType) النوع البياني للمصفوفة، ويمكن الاستغناء عنها.

مثال:

يتم الاعلان عن مصفوفة خالية

Dim ArrName() As Double

ثم يتم تحديد عدد العناصر (المرّة الأولى) كالتالي:

ReDim ArrName (49) As Double

ArrName (0) =25.50

ArrName (1) =12.75

.

.

.

ArrName (49) =40.30

ثم يتم إعادة تحديد عدد العناصر (المررة الثانية) كالتالي:

لو أردنا أن نضيف عنصر واحد إلى هذه المصفوفة مع الاحتفاظ بالقيم السابقة
نستخدم عبارة Preserve كالتالي:

ReDim Preserve ArrName(50) As Double

سيتم الإبقاء على القيم السابقة، مضافا إليها عنصر واحد فارغ جديد هو ArrName(50)
ويتم استخدامه كالتالي:

ArrName(50)=32.10

ملاحظة مهمة جدا: إن الأمر Redim لا يستخدم إلا داخل الإجراء فقط
(أي لا يستخدم في قسم الإعلانات مثل الأمر Dim)

ثانيا :**المصفوفات متعددة الأبعاد MultiDimensional Arrays****مقدمة:****ماهي المصفوفات متعددة الابعاد؟**

بعض البيانات يمكن وضعها في مصفوفة ذات بعد واحد كما ذكرنا سابقا وهناك بعض البيانات لا يكون من المناسب استخدام المصفوفات ذات البعد الواحد بل يجب استخدام المصفوفات متعددة الأبعاد.

فمثلا لو ان لدينا بيانات لمجموعة من الطلاب

(الاسم - المادة - الدرجة النصفية - الدرجة النهائية)

ونريد إيجاد مجموع الدرجات والتقدير لجميع الطلاب ففي مثل هذه الحالة يجب استخدام المصفوفات متعددة الأبعاد.

وطريقة التعامل معها لا تختلف كثيرا عن التعامل مع المصفوفات ذات البعد الواحد ولكن الفرق الجوهرى هو في عدد فهارس المصفوفة.

ببساطة عندما تريد انشاء مصفوفة .. ضع في اعتبارك ان عدد الفهارس هو نفس عدد الابعاد للمصفوفة.

كما سوف نرى لاحقا.

والمصفوفة متعددة الأبعاد تشمل أي مصفوفة لها أكثر من بعد واحد.

مثال :

المصفوفة الثنائية تشبه الجدول تماما و تحتوي على بعدين هما الطول والعرض وبالتالي تحتاج لرقمي فهرس.

ولكي يتم الوصول إلى أي عنصر فيها نستخدم دليلين معا (فهرسين 2 Index)

الدليل الأول يكون لرقم العمود ،

والدليل الثاني لرقم الصف.

مثال آخر :

المصفوفة ثلاثية الأبعاد , وهي تشبه المكعب حيث تحتوي على ثلاثة أبعاد هي الطول والعرض والارتفاع وبالتالي تحتاج لثلاثة فهارس.

وهكذا...

يتيح لك Visual Basic أن تقوم بتعريف مصفوفات يصل حجمها حتى 60 بعدا !!

وفي الغالب لن تستخدم كل هذه الأبعاد.

واهم ما في المصفوفات متعددة الأبعاد هي المصفوفة ذات البعدين (المشابهة للجدول) وهي كافية تماما لمعظم الأغراض البرمجية.

وعلى أقصى تقدير لا أظن أنك ستحتاج إلى مصفوفة لها أكثر من ثلاثة أبعاد.

طرق الإعلان :

هناك ثلاثة طرق للإعلان عن مصفوفة متعددة الأبعاد

1- بتحديد القيمة القصوى لكل بعد:

Dim ArrayName(MaxCols,MaxRows) As DataType

كما في المصفوفة ذات البعد الواحد نبدأ

بنوع مدى رؤية المصفوفة Dim او Public او Private

ثم اسم المصفوفة ArrayName

ثم حجم المصفوفة

وهنا تم وضع بعدين فقط هما MaxCols و MaxRows

إذن هذه المصفوفة ثنائية البعد

MaxCols : تمثل البعد الاول وهو الطول الأقصى للعمود (الحقل)

MaxRows: تمثل البعد الثاني وهو الطول الأقصى للعرض (للسجل)

وننتهي بـ DataType وهي تمثل النوع البياني للمصفوفة

مثال (لمصفوفة ثنائية) :**Dim XArray(3, 5) As STRING****حيث:****XArray: هو اسم المصفوفة****3 : وهي تمثل عدد أعمدة المصفوفة (4 حقول)****5 : وهي تمثل عدد صفوف المصفوفة (6 سجلات)****STRING: نوع المصفوفة وهي مصفوفة نصية.****ويكون مجموع العناصر (الخلايا) داخل المصفوفة الثنائية هو 24 عنصرا ،****أي حاصل ضرب البعد الأول (4 حقول)****في البعد الثاني (6 سجلات)****4 * 6 = 24 عنصرا (خلية)**

مثال (لمصفوفة ثلاثية الأبعاد) :

وبما أن المصفوفة ثلاثية الأبعاد فإنها تحتوي على الطول – العرض – الارتفاع وبالتالي كل عنصر يحتوي على ثلاثة فهارس لتمييزه عن العناصر الأخرى داخل المصفوفة

Dim XArray (8, 10, 12) as Double

حيث:

XArray: هو اسم المصفوفة

8 : وهي تمثل طول المصفوفة (9 عناصر)

10 : وهي تمثل عرض المصفوفة (11 عنصر)

12 : وهي تمثل ارتفاع المصفوفة (13 عنصر)

Double: نوع المصفوفة وهي مصفوفة نصية.

ويكون مجموع العناصر (الخلايا) داخل المصفوفة الثلاثية

هو 1287 عنصرا ،

أي حاصل ضرب الطول (9 عناصر)

في العرض (11 عنصر)

في الارتفاع (13 عنصر)

$$1287 = 13 * 11 * 9 \text{ عنصرا (خلية)}$$

2- بتحديد القيمة البداية والنهاية لكل بعد:**Dim ArrayName(MinCols to MaxCols, MinRows to MaxRows) As DataType****DIM** مجال الرؤية للمصفوفة محدود**ArrayName** اسم المصفوفة**MinCols** بداية البعد الأول للمصفوفة**MaxCols** نهاية البعد الأول للمصفوفة**MinRows** بداية البعد الثاني للمصفوفة**MaxRows** نهاية البعد الثاني للمصفوفة**DataType** النوع البياني للمصفوفة**مثال (لمصفوفة ثنائية) :****Dim XArray(2 TO 4, 4 TO 8) As STRING****البعد الأول:**

يبدأ من 2 إلى 4 أي تحتوي على 3 حقول.

البعد الثاني:

يبدأ من 4 إلى 8 أي تحتوي على 5 سجلات.

ويكون مجموع العناصر (الخلايا) داخل المصفوفة الثنائية هو 25 عنصرا ،

أي حاصل ضرب البعد الأول (3 حقول)

في البعد الثاني (5 سجلات)

 $3 * 5 = 25$ عنصرا (خلية)

3- يمكن الإعلان عنها بدون تحديد عناصر (المصفوفة الخالية من العناصر).**مثال:****Dim ArrName() As DataType**

وفي هذه الحالة تكون المصفوفة فارغة تماما ولا يمكن التعامل معها إلا بعد إعادة تعريفها عن طريق الأمر ReDim الذي تم شرحه بالتفصيل. وهذه الطريقة لا تختلف عن المصفوفات ذات البعد الواحد إلا بعد إعادة التعريف فان تم إعادة تعريفها ببعد واحد أصبحت مصفوفة أحادية البعد , وان تم إعادة تعريفها بأكثر من بعد أصبحت مصفوفة متعددة الأبعاد.

مثال:**Dim ArrName() As String****ReDim ArrName(2,2)**

وبعد أن قمنا بإعادة تعريف المصفوفة ArrName نقول الآن أنها مصفوفة ثنائية الأبعاد طولها 3 حقول وعرضها 3 سجلات

التعامل مع المصفوفة متعددة الأبعاد:**مثال:**

لو كان هنا 50 طالبا ونريد تخزين أسماء الطلاب والدرجة النهائية لكل طالب فإننا نحتاج لمصفوفة تتكون من 50 عنصرا للأسماء و 50 عنصرا للدرجات.. فيكون الإعلان من النوع ثنائي البعد كالتالي:

Dim Stud(1,49) As String

- اسم المصفوفة Stud

- طولها 100 عنصرا , لان المصفوفة تحتوي على 2 حقل و 50 سجل
إذن $100 = 50 * 2$ عنصر (خلية)

- النوع البياني للمصفوفة هو String

وطريقة التعامل معها كالتالي:**أولا: إضافة البيانات**

Stud(0,0) = "Ahmad"
Stud(1,0) = 88

Stud(0,1) = "Ali"
Stud(1,1) = 79

Stud(0,2) = "Majed "
Stud(1,2) = 100

.

.

ثانيا الحصول على البيانات:

Text1.Text= Stud(0,0)
Text2.Text= Stud(1,0)
Text3.Text= Stud(0,1)

.

.

.

ملاحظة مهمة:

المصفوفات المتعددة الأبعاد تشبه أحادية البعد من حيث الاستخدام، إذ ينطبق عليها عبارة Option Base

كذلك ينطبق على هذا النوع استخدام العبارة ReDim , إذ يمكن تغيير الأبعاد كأي مصفوفة أحادية البعد.

الأمر المهم هنا أنك لو استخدمت Preserve مع المصفوفة متعددة الأبعاد، فإن البعد الأخير فقط هو الذي يمكن تغيير عدد عناصره.

مثال:

Dim XArray() as Integer

ReDim XArray(2,3)

هنا في هذه الحالة لا يمكن تغيير البعد الأول ولكن لك الحرية في تغيير طول البعد الأخير وهو هنا البعد الثاني كما يلي:

ReDim Preserve XArray(2,8)

وإذا قمنا باعادة التعريف بتغيير طول البعد الأول

ReDim Preserve XArray(3,3)

فان البرنامج يصدر رسالة خطأ.

مهارت في التعامل مع المصفوفات

اولا :

استخدام دالة التكرار For

تستخدم دالة التكرار For في التعامل مع المصفوفات لكي يتم المرور على جميع عناصر المصفوفة , فمثلا لو أردنا عرض درجات مائة طالب فإننا نحتاج لكود يتكون من مائة سطر ولكن مع استخدام دوال التكرار فقد لا يتجاوز طول الكود عن الثلاثة أسطر

الصيغة العامة لـ FOR :

FOR I = Xstart TO Xend

NEXT I

FOR : اسم دالة التكرار

I : يمثل متغير التكرار

Xstart : تمثل قيمة بداية التكرار

Xend : تمثل قيمة نهاية التكرار

NEXT I : إعادة مؤشر القراءة لبداية التكرار مع زيادة متغير التكرار

الشرح :

تعمل هذه الدالة على تكرار الأكواد الواقع بين FOR و NEXT ، بعدد معين من المرات يتم تحديد عدد مرات التكرار عن طريق قيمة البداية Xstart وقيمة النهاية Xend للمتغير (I) ، وفي كل مرة يتم زيادة المتغير (I) بواحد إلى أن يصبح أكبر من قيمة النهاية Xend عندها يتوقف التكرار.

ملاحظة :

يمكن وضع شرط داخل دالة التكرار بحيث إذا تحقق الشرط يتم الخروج من التكرار حتى ولو لم يصل متغير التكرار إلى نهاية التكرار.

مثال:

```

FOR I = 1 TO 6
IF I = 4 THEN
EXIT FOR
END IF
NEXT I

```

ملاحظة :

يتم زيادة متغير التكرار بواحد افتراضيا كلما يتم تنفيذ التكرار على الاكواد الخاصة بالدالة ولكن لو أردت زيادة متغير التكرار بـ 4 أو 2 أو 3 أو فيمكنك ذلك عن طريق الأمر STEP

مثال:

لو أردنا أن نقوم بعرض الأرقام الزوجية من الصفر إلى 10 فيكون الكود كالتالي

```

FOR I = 0 TO 10 STEP 2
MsgBox I
NEXT I

```

في المثال السابق الأعداد التي سوف يتم عرضها هي 0 , 2 , 4 , 6 , 8 , 10 فقط لأنه يتم زيادة متغير التكرار بالرقم 2 بدلا من الواحد

ويمكن أن يكون التغيير على متغير التكرار بالنقصان كالتالي:

```

FOR I = 10 TO 0 STEP -2
MsgBox I
NEXT I

```

في المثال السابق الأعداد التي سوف يتم عرضها هي 10 , 8 , 6 , 4 , 2 , 0

كيف يمكن استخدام دالة التكرار مع المصفوفات؟

ببساطة أي مصفوفة لكي يتم التعامل مع عناصرها يتم ذلك بتحديد رقم الفهرس INDEX للعنصر وبدلاً من أن تحدد هذا الرقم بنفسك فإن دالة التكرار تحدد بالنيابة عنك.

مثال :

نريد أن نخزن في مصفوفة الأعداد من 1 إلى 100

```
Dim Xarr(1 To 100) As Integer
```

```
For I = 1 To 100
```

```
  Xarr(I) = I
```

```
Next I
```

هل رأيت كيف تم اختصار هذا المهمة عن طريق دالة التكرار؟؟؟

ثانياً :**معرفة طول المصفوفة**

يمكنك معرفة عدد عناصر مصفوفة باستخدام الدالة Ubound

الصيغة العامة

```
Ubound ( NAME ARRAY)
```

حيث NAME ARRAY هي اسم المصفوفة

مثال:

```
Dim Xarr(1 To 100) As Integer
```

```
MsgBox UBound(Xarr)
```

النتيجة سوف يكون 100

ثالثا :**كيف يمكن التعامل مع بيانات مصفوفة بعدة أنواع من البيانات؟؟؟**

عندما يتم إنشاء مصفوفة فإنها تأخذ نوع بياني واحد ولكن هناك طريقة للتغلب على هذه المشكلة وهي كالتالي :

يتم الإعلان عن مصفوفة من نوع String أو Variant وهذان النوعان يقبلان جميع الأنواع البانية الأخرى ولكن عندما نريد التعامل مع عناصرها وإجراء العمليات الحسابية فإنه يجب استخدام دوال التحويل البانية

وهي :

Integer للتحويل إلى CInt
 Boolean للتحويل إلى CBool
 Byte للتحويل إلى CByte
 Currency للتحويل إلى CCur
 Data للتحويل إلى CDate
 Double للتحويل إلى CDbl
 Single للتحويل إلى CSng
 String للتحويل إلى CStr
 Variant للتحويل إلى CVar

مثال:

```
Dim a As String
Dim b As String
```

```
a = 10
```

```
b = 20
```

```
MsgBox a + b
```

```
MsgBox CInt(a) + CInt(b)
```

في الرسالة الأولى لن يتم جمع العددين ولكن سوف يتم دمجهم لأنهم في صورة نصية فيكون الناتج كالتالي 1020

أما الرسالة الثانية وبعد استخدام دالة التحويل فإن المتغيرات النصية سوف تحول إلى متغيرات عددية وبالتالي سوف يتم جمعها فيكون الناتج كالتالي 30

مثال آخر:

Dim Xarr(2) As String

Xarr(1) = "01/01/2005"

مقارنة التاريخ في المصفوفة بالتاريخ الحالي'

If CDate(Xarr(1)) < Date Then

MsgBox "هذا التاريخ قديم"

End If

أمثلة على التعامل مع المصفوفات

المثال الأول :

يقوم هذا المثال بتخزين بيانات داخل مصفوفة ثم يقوم بترتيبها ومن ثم عرضها

الفكرة بسيطة جدا

نفترض أننا قمنا بالإعلان عن المصفوفة كالتالي

Dim Arr(5) As Integer

مصفوفة رقمية تحتوي على 6 عناصر

وقمنا بعد ذلك بوضع بتعبئة عناصر البيانات كالتالي

Arr(0) = 8011

Arr(1) = 2055

Arr(2) = 4066

Arr(3) = 3044

Arr(4) = 6088

Arr(5) = 1000

الان جاء دور ترتيب البيانات عناصر المصفوفة لكي نقوم بذلك يجب ان نقوم باخذ العنصر الاول في المصفوفة ومقارنته مع باقي العناصر فان وجد عدد اقل من العدد الاول يقوم بعملية الإبدال بين العنصر الاول ، والعنصر الذي قيمته اصغر من العنصر الاول ، وطبعا هنا نحتاج لمتغير للاحتفاظ بقيمة العنصر الاول ، ومن ثم اسناد قيم العنصر الاصغر للعنصر الاول في المصفوفة ، وقيمة العنصر الاول التي وضعناها في المتغير نقوم باسنادها للعنصر الاصغر ، ثم بعد ذلك ينتقل للعنصر الثاني ويكرر نفس الطريقة السابقة .. وهكذا

حتى الانتهاء من المرور على جميع عناصر المصفوفة

طبعا نحتاج لعبارتي تكرار ، فيكون الكود كالتالي

'المتغير الخاص بالاحتفاظ بقيمة العنصر الأصغر لعملية الإبدال

Dim X_Var As Integer

'اللوب الخاص بالمرور على العناصر التي سوف نقارنها
'بباقي العناصر

For I = 0 To 5

'اللوب الخاص بمقارنة العنصر المحدد بباقي العناصر

For z = 0 To 5

'هنا شرط الإبدال أن يكون العنصر الأصغر في الأعلى

If Arr(I) < Arr(z) Then

'عملية الإبدال

**X_Var = Arr(I)
Arr(I) = Arr(z)
Arr(z) = X_Var**

End If

Next z

Next I

ومن ثم عرضها في ListBox كالتالي:

**For I = 0 To 5
List2.List(I) = Arr(I)
Next I**

ويوجد مثال مرفق للتوضيح

UnderTiker_Arr1.rar

المثال الثاني:

يقوم هذا المثال باخذ بيانات الطلاب من جدول في قاعدة البيانات ووضعها في مصفوفة متعددة الابعاد (ثنائية) ومن ثم نقوم بإيجاد إجمالي الدرجات للطلاب والتقدير لكل طالب

يحتوي الجدول على الحقول التالية :

رقم الطالب
اسم الطالب
الدرجة الشهرية (من 10 درجات)
الدرجة النصفية (من 30 درجة)
الدرجة النهائية (من 60 درجة)

نقوم بإيجاد الإجمالي (الشهري + النصفى + النهائي)

وتخزينها في حقل خاص في المصفوفة ، ويكون الحقل الاخير للتقدير
بناء على المعادلة التالية

1- ممتاز : اذا كان المجموع اكبر من او يساوي 90

2- جيد جدا : اذا كان المجموع اكبر من او يساوي 80

3- جيد: اذا كان المجموع اكبر من او يساوي 70

4- ضعيف : اقل من ذلك

أولا: نعلن عن مصفوفة خالية من العناصر من نوع نصي

Dim X_Arr() As String

ثانيا: نقوم باعادة تعريفها

ولكن كم نحتاج من الحقول والسجلات ؟؟؟

بالنسبة للحقول فالذي نحتاجه 7 حقول

هي (الرقم - الاسم - الدرجة الشهرية - النصفية - النهائية - المجموع - التقدير)

وبالنسبة لعدد السجلات فنحصل عليه من خلال مؤشر السجلات

عن طريق الخاصية RecordCount

فيكون كود إعادة التعريف كالتالي

```
Dim X_Count As Integer
```

```
X_Count = RecSet.RecordCount
```

```
ReDim X_Arr(1 To 7, 1 To X_Count) As String
```

ثالثا: نقوم بتعبئة بيانات الجدول في المصفوفة كالتالي:

```
For i = 1 To X_Count
```

```
    If RecSet.EOF Then Exit For
```

```
    X_Arr(1, i) = RecSet.Fields.Item("No_Stud").Value
```

```
    X_Arr(2, i) = RecSet.Fields.Item("Name").Value
```

```
    X_Arr(3, i) = RecSet.Fields.Item("Mark_Month").Value
```

```
    X_Arr(4, i) = RecSet.Fields.Item("Mark_Half").Value
```

```
    X_Arr(5, i) = RecSet.Fields.Item("Mark_Final").Value
```

```
    RecSet.MoveNext
```

```
Next i
```

رابعاً: نقوم بعمليات المعالجة (المجموع - التقدير)**كالتالي:**

'متغير لتخزين اجمالي الدرجات للطالب

Dim X_Result As Double**For i = 1 To X_Count**

'الحصول على مجموع الدرجات للطالب

X_Result = 0**X_Result = Cdbl(X_Arr(3, i)) + Cdbl(X_Arr(4, i)) + Cdbl(X_Arr(5, i))****X_Arr(6, i) = X_Result**

'الحصول على التقدير

Select Case X_Result**Case Is >= 90****X_Arr(7, i) = "ممتاز"****Case Is >= 80****X_Arr(7, i) = "جيد جداً"****Case Is >= 70****X_Arr(7, i) = "جيد"****Case Else****X_Arr(7, i) = "ضعيف"****End Select****Next i**

أخيرا تبقى لنا عرض المصفوفة اداة الشبكة MSFlexGrid1

'اللوب الخاص بالتنقل بين السجلات

For i = 1 To X_Count

'تحديد سجل الادخال في الشبكة

MSFlexGrid1.Row = MSFlexGrid1.Rows - 1

'اللوب الخاص بالتنقل بين حقول المصفوفة والشبكة

For z = 0 To 6

'تحديد رقم الحقل في الشبكة

MSFlexGrid1.Col = z

'تحديد القيمة من المصفوفة واسنادها للحقل في الشبكة

MSFlexGrid1.Text = X_Arr(z + 1, i(

Next z

'زيادة عدد صفوف الشبكة

MSFlexGrid1.Rows = MSFlexGrid1.Rows + 1

Next i

وقبل هذا نحتاج لتهيئة الشبكة

وللمعلومة أضع كود التهيئة هنا

Private Sub DatGrid()

'مسح الشبكة

MSFlexGrid1.Clear

'تحديد عدد الحقول

MSFlexGrid1.Cols = 7

'تحديد عدد السجلات

MSFlexGrid1.Rows = 2

'تحديد عدد سجلات العناوين

MSFlexGrid1.FixedRows = 1

'تحديد عدد حقول العناوين

MSFlexGrid1.FixedCols = 0

'تسمية صف العناوين

MSFlexGrid1.Row = 0

MSFlexGrid1.Col = 0

MSFlexGrid1.Text = "رقم الطالب"

MSFlexGrid1.Col = 1

MSFlexGrid1.Text = "الاسم"

MSFlexGrid1.Col = 2

MSFlexGrid1.Text = "الشهري"

MSFlexGrid1.Col = 3

MSFlexGrid1.Text = "النصفي"

MSFlexGrid1.Col = 4

MSFlexGrid1.Text = "النهائي"

MSFlexGrid1.Col = 5

MSFlexGrid1.Text = "المجموع"

MSFlexGrid1.Col = 6

MSFlexGrid1.Text = "التقدير"

'تنظيم عرض الحقول

```
MSFlexGrid1.ColWidth(0) = 1000  
MSFlexGrid1.ColWidth(1) = 2000  
MSFlexGrid1.ColWidth(2) = 1000  
MSFlexGrid1.ColWidth(3) = 1000  
MSFlexGrid1.ColWidth(4) = 1000  
MSFlexGrid1.ColWidth(5) = 1000  
MSFlexGrid1.ColWidth(6) = 1000
```

'تنسيق الخلايا

```
For i = 0 To 6  
    MSFlexGrid1.ColAlignment(i) = 3  
    MSFlexGrid1.FixedAlignment(i) = 3  
Next i
```

End Sub

يوجد مثال مرفق للتوضيح

UnderTiker_Arr2.rar

اسئلة الكتاب

السؤال الأول:

احب بصح أو خطأ مع التعليل في حال الإجابة خطأ

- 1- تتكون المصفوفة من مجموعة من الفهارس Index
- 2- يمكن أن تتكون المصفوفة من عدة أنواع بيانية.
- 3- يتميز كل عنصر داخل المصفوفة برقم Index فريد.
- 4- هناك نوعان للمصفوفة مصفوفة ذات بعد واحد ومصفوفة متعددة الأبعاد حيث المصفوفة ذات البعد الواحد تشبه الجدول.
- 5- المصفوفة طولها ثابت ولا يمكن زيادته أثناء عمل البرنامج.
- 6- يبدأ الإعلان عن المصفوفة بتحديد النوع البياني للمصفوفة.
- 7- عند الإعلان عن مصفوفة Public فإنه يتم استخدامها داخل الإجراء فقط.
- 8- Dim Arr(100) As String
تتكون هذه المصفوفة من 100 عنصر من نوع String .
- 9- يتم استخدام Option Base لتحديد طول المصفوفة.
- 10- يعتبر الإعلان عن المصفوفة بطريقة البداية والنهاية أكفاً من استخدام عبارة Option Base في تحديد رقم البداية للمصفوفة.
- 11- يمكن أن نبدأ الترقيم للمصفوفة بأي عدد نريده عن طريق Option Base
- 12- Dim Arr(3 TO 4) As Integer
تتكون هذه المصفوفة من أربعة عناصر من نوع Integer
- 13- لا يمكن الإعلان عن مصفوفة خالية.
- 14- الدالة Array تقوم بتهيئة مصفوفة من أي نوع بياني.
- 15- يمكن التعامل مع المصفوفة الخالية من العناصر وذلك بعد إعادة تعريفها عن طريق الدالة Array
- 16- عند إعادة تعريف مصفوفة فإن Preserve تعتبر كلمة اختيارية تقوم بحفظ طول المصفوفة.
- 17- للتعامل مع المصفوفات الديناميكية يتم الإعلان عن المصفوفة أولاً بإحدى طرق الإعلان عنها ومن ثم يتم إعادة تعريفها عن طريق ReDim
- 18- عدد الفهارس في المصفوفة مساو لعدد عناصرها.
- 19- المصفوفة متعددة الأبعاد تشمل أي مصفوفة لها أكثر عنصر.
- 20- يتيح لك الفيچوال أن تقوم بتعريف مصفوفات يصل حجمها 60 بعداً .
- 21- Dim XArray(3 to 5) As STRING
الإعلان السابق يمثل مصفوفة ثنائية الأبعاد

Dim XArray(4, 3) As STRING-22
تكون المصفوفة السابقة من 12 عنصرا.

23- يمكننا تغيير عدد عناصر أي بعد للمصفوفة متعددة الابعاد
مع الاحتفاظ بالقيم عن طريق **ReDim Preserve**

السؤال الثاني:

اكتشف الأخطاء في هذا الكود

Dim XArray (2,8) As String
Dim XArr2 () As String

XArray (3,1) = Ahmad
XArray (4,2) = Ali
XArr2 (2) = 120

Text1 = XArray

الإجابات

إجابة السؤال الأول

- 1- خطأ ... تتكون المصفوفة من مجموعة من العناصر .
- 2- خطأ ... لا يمكن أن تتكون المصفوفة من عدة أنواع بيانية.
- 3- صح ... يتميز كل عنصر داخل المصفوفة برقم Index فريد.
- 4- خطأ ... هناك نوعان للمصفوفة مصفوفة ذات بعد واحد ومصفوفة متعددة الأبعاد حيث المصفوفة ذات البعد الواحد تشبه السجل.
- 5- خطأ ... المصفوفة طولها غير ثابت و يمكن زيادته أثناء عمل البرنامج.
- 6- خطأ ... يبدأ الإعلان عن المصفوفة بتحديد نوع المدى (المجال SCOPE) للمصفوفة.
- 7- خطأ ... عند الإعلان عن مصفوفة Public فانه يتم استخدامها من أي مكان داخل البرنامج.
- 8- Dim Arr(100) As String
خطأ ... تتكون المصفوفة من 101 عنصر من نوع String , لان الترقيم يبدأ من الصفر مالم يتم الإعلان عن Option Base 1.
- 9- خطأ ... يتم استخدام Option Base لتحديد رقم البداية للمصفوفة , عند الإعلان عن Option Base فانه يمكن بدء الترقيم من الواحد أو الصفر فقط.
- 10- صح ... يعتبر الإعلان عن المصفوفة بطريقة البداية والنهاية أكفاً من استخدام عبارة Option Base في تحديد رقم البداية للمصفوفة.
- 11- خطأ ... يمكن أن نبدأ الترقيم للمصفوفة من الواحد أو الصفر فقط عن طريق Option Base.
- 12- Dim Arr(3 TO 4) As Integer
خطأ ... تتكون هذه المصفوفة من عنصرين فقط من نوع Integer
- 13- خطأ ... يمكن الإعلان عن مصفوفة خالية, و لكن لا يمكن التعامل معها إلا بعد إعادة تعريفها عن طريق ReDim
- 14- خطأ ... الدالة Array تقوم بتهيئة مصفوفة من نوع Variant فقط.
- 15- خطأ ... يمكن التعامل مع المصفوفة الخالية من العناصر وذلك بعد إعادة تعريفها عن طريق الامر ReDim.
- 16- خطأ ... عند إعادة تعريف مصفوفة فان Preserve تعتبر كلمة اختيارية تقوم بالاحتفاظ بالعناصر الموجودة حالياً في المصفوفة.
- 17- خطأ ... للتعامل مع المصفوفات الديناميكية يتم الإعلان عن المصفوفة الخالية من العناصر ومن ثم يتم إعادة تعريفها عن طريق ReDim.
- 18- خطأ ... عدد الفهارس في المصفوفة مساو لعدد ابعادها.
- 19- خطأ ... المصفوفة متعددة الابعاد تشمل أي مصفوفة لها اكثر من بعد.
- 20- صح ... يتيح لك الفيچوال أن تقوم بتعريف مصفوفات يصل حجمها 60 بعداً .

(مشرفة منتدي الفيڤوال بيسك بمندقيات المحيط العربي)

Dim XArray(3 to 5) As STRING -21خطا ... الاعلان السابق يمثل مصفوفة إحادية البعد.**Dim XArray(4, 3) As STRING -22**خطا ... تتكون المصفوفة السابقة من 20 عنصرا.**-24 خطا ... يمكننا تغيير عدد عناصر البعد الاخير للمصفوفة فقط في المصفوفة متعددة الابعاد مع الاحتفاظ بالقيم عن طريق ReDim Preserve.****إجابة السؤال الثاني :**

اكتشف الأخطاء في هذا الكود

Dim XArray (2,8) As String

عبارة صحيحة

Dim XArr2 () As String

عبارة صحيحة

XArray (3,1) = Ahmad

1- يجب استخدام علامات التنصيص " " عند إسناد النصوص
 2- البعد الأول خارج نطاق المصفوفة .. لان هذا البعد ينتهي عند الرقم 2 .. والمستخدم هو الرقم 3

XArray (4,2) = Ali

1- يجب استخدام علامات التنصيص " " عند إسناد النصوص
 2- البعد الأول خارج نطاق المصفوفة .. لان هذا البعد ينتهي عند الرقم 2 .. والمستخدم هو الرقم 4

XArr2 (2) = 120

لا يمكن استخدام مصفوفة خالية إلا بعد إعادة تعريفها
 و المصفوفة XArr2 لم يتم إعادة تعريفها.

Text1 = XArray

الخطأ هنا هو عدم تحديد رقم الفهرس للعنصر المطلوب إسناده إلى Text1.

تم بحمد الله

ارجوا أن لا تنسوني من صالح الدعاء

مع تحيات :

أخوكم : إياد عماد نوفل

الاندريتيكر